
ABSTRACT 

We exemplify how SAS
®
 can be used to perform analyses of genetic data. In particular, we perform a population 

genetic analysis based on a dataset containing microsatellite data in a population of Plasmodium falciparum (the 
parasite that causes the most virulent form of human malaria).  The microsatellite markers are located around the 
dhps gene of P. falciparum, which is associated with resistance against the antimalarial drug Sulfadoxine. 
Resistance-causing mutations are selectively favored and spread in the population. A selective sweep, or the 
“hitchhiking” effect of an advantageous mutation, refers to the elimination of pre-existing genetic variation. Studying 
the pattern of genetic variation is important because by measuring the span of selective sweeps we may infer the 
selective pressures on drug resistance and relate it to the demographic and clinical factors specific to the population 
under examination. We illustrate how various components of SAS

®
 can be used and linked together to perform the 

desired analysis.  

INTRODUCTION 

AIM 

We exemplify how to use SAS
®
 to perform a simple analysis of genetic data. We choose an example that is relevant 

for malaria control. The aim of the manuscript is to illustrate how to perform meaningful analyses and statistics 
starting from a dataset (such as an EXCEL spreadsheet) that was created in a naïve way.  

WHAT IS MALARIA? 

Malaria is still a threat to the public health in large areas of the developing world. It is an infectious disease produced 
by (unicellular eukaryotic) protozoan parasites of the genus Plasmodium, with Plasmodium falciparium being 
responsible for the most virulent form of human malaria (cf. Snow et al., 2005). P. falciparum causes high morbidity 
and mortality, which annually results in 200 million to 300 million infections and one to three million deaths (cf. WHO 
2000). Malaria is a vector-borne disease with a complex transmission cycle having sexual phases in the mosquito 
vector and asexual phases in the infected host (e.g., Daily, 2006).   

DRUG-RESISTANCE IN MALARIA 

Many effective and widely used drugs, e.g., Chloroquine or Sulfadoxine and Pyrimethamine, have been rendered 

useless because parasites rapidly evolved resistance against them. The limited repertoire of safe, effective, and 
affordable anti-malarial drugs has made research on the emergence and dispersion of resistance a global health 
priority (see Marsh, 1998). 

 

IDEA BEHIND THE GENETIC ANALYSIS 

We need to reconstruct the past dynamics of drug resistance in a given population to understand which factors lead 
to the spread of resistant parasites. In the absence of reliable public-health records, such retrospective analysis may 
not be feasible. However, fast accumulation of parasite genome-sequence data provides a means to examine 
indirectly the past events of drug-resistance evolution without epidemiological data: by examining “selective sweeps” 
around the loci of drug resistant mutations (cf. Schneider and Kim 2010).  

A selective sweep, or the “hitchhiking” effect of an advantageous mutation, refers to the elimination of pre-existing 
genetic variation when a particular chromosome segment carrying a favored allele sweeps through the population 
(Maynard Smith and Haigh, 1974; Stephan et al., 1992; Barton, 2000). The extent of this wipeout depends on how 
fast the favored allele increases to high frequency while meiotic recombination is constantly eroding the association 
between the favored allele and the surrounding chromosome segment. The chromosomal span of reduced variation 
thus depends on the relative reproductive advantage of resistant alleles over sensitive alleles, which determines the 
speed of frequency increase. Therefore, by measuring the span of selective sweeps, we may infer the selective 
pressures on drug resistance and relate it to the demographic and clinical factors specific to the population under 
examination.  

Label Dhfr MS locus 1 MS locus 3 MS locus 3 MS locus 4 

F70 3 307  101 290 

 3 314   286 

 3    278 

 3    296 

Table 1. The table shows the entry for sample “F70”. The parasites have allele A3 at the dhps locus. At the 
first MS locus, alleles with 307 and 314 repeats of the core sequence were found. The entry for MS locus 2 is 
empty. At MS locus 3 only alleles with 101 repeats of the cores sequence were detected, and at MS locus 4, 
four different alleles were found in sample „F70‟. 

DATA AND ANALYSIS 

For our illustration purposes, we use a simulated dataset that imitates an original dataset obtained from a study in 
Kenya.   

Blood samples were taken from patients infected with malaria caused by P. falciparum. The parasites were extracted 
from the bold, and parts of the genome were sequenced. In particular, a gene associated with resistance against 
Sulfadoxine (dhps gene) was sequenced. Moreover, for both genes several microsatellite (MS) markers were 
sequenced. Microsatellites are highly repetitive DNA sequences, which are typically neutral.  

The dataset contains the following information. One or more consecutive rows represent the genetic variation from 
the parasites of one blood sample (Table 1 shoes an example). The first column labels the blood sample. If more 
than on rows represent the same blood sample, the label is specified only in the respective first row. Column three 
specifies the particular allele found at the dhfr locus associated with drug resistance. Again entered only in the first 
row of a given sample (only such samples were included in which all parasites had the same allele at the dhfr locus). 
The remaining columns represent the microsatellite variation at different genome positions. For a given blood sample 
the entries in the rows, are the number of sequence repeats found at the various microsatellite loci in a given blood 
sample. Figure 1 shows a screenshot of the initial dataset in .xls format.    

  

 

Figure 1. Screen shot of the EXCEL spreadsheet.  

IMPORTING THE DATASET  

First, we import the EXCEL spreadsheet into SAS
®
. This can be done by using the import wizard or by the following 

code: 

PROC IMPORT OUT= sasgf.dhps1  

            DATAFILE="C:\Documents and 

Settings\User\SASGF2011\SASGF2011SampleData.xls"  

            DBMS=EXCEL REPLACE; 

     RANGE="microsatellites$";  

     GETNAMES=YES; * column names created from first row; 

     MIXED=NO; 

     SCANTEXT=YES; 

     USEDATE=YES; 

     SCANTIME=YES; 

RUN; 

 

This code stores the spreadsheet as „dhps‟ in the SAS
®
 library „SASGF‟ (see Figure 2 for an illustration of the 

imported data).  

 

                

Figure 2. Screenshot of dataset „sasgf.dhps1‟ (left), and „work.dhps1‟ after performing the two data steps 
described in step 1. The first column „sample‟ was deleted, and the 0-1 variable „sample1‟ was added.  

CALCULATING MEASURES FOR GENETIC VARIABILITY 

We are interested in the genetic variability at each MS locus. A measure for the genetic variability is the 
heterozygosity. It is calculated as 

  
 

   
(  ∑  

 

 

   

)  

For a given MS locus n refers to the number of different alleles found among all samples at this locus. Moreover, pi is 
the relative frequency of MS allele Ai (calculated among all samples). We want to calculate the heterozygosity 
separately for the blood samples that contain resistant and sensitive parasites, respectively.  

For a given MS locus, some blood samples contain various MS alleles. When calculating the relative frequencies of 
the MS alleles we have to take this into account. Therefore, for each blood sample, we weigh the MS alleles 
reciprocal to the number of alleles found in this sample. In the example of Table 1, we weigh the alleles at MS locus 1 
by ½, that of MS locus 3 by 1, and those of MS locus 4 by 1/4. At MS locus 2 the entries are missing, so they are not 
counted at all.  

 

Step1: Manipulating the data 

Within the sample there are various alleles at the dhps locus, A1, A2, ... . Parasites carrying the allele A2 are resistant 
whereas all others are sensitive. Since we want to distinguish only between sensitive and resistant, we replace the 

column allele by a 0-1 variable, where '1' codes for resistant and '0' for sensitive. Entries with missing value for 'allele' 
are deleted since they are uninformative. This is done with a basic data step: 

data work.dhps1; 

set sasgf.dhps1; 

if allele=0 then allele=0; 

if allele=1 then allele=0; 

if allele=2 then allele=1; 

if allele=3 then allele=0; 

if allele=. then delete; 

run; 

 

The first problem that arises with our data is that not every row contains a label referring to the respective blood 
sample. The second problem is that the labels are not in numeric format.  We overcome the two problems in two 
steps.  

We add a column 'sample1' to „SASGF.dhps‟, with an entry „1‟ if a row contains a label for the sample, and „0‟ if it 
does not contain a label. Moreover, we delete the column 'sample'. 

data work.dhps1; 

set dhps1; 

if sample=' ' then sample1=0; 

else sample1=1; 

drop sample; 

run; 

 

The output is illustrated in Figure 2Figure 3.  

Hypothetically, one can overwrite the column 'sample' instead of deleting it and creating a new one. However, the 
column 'sample' has character format. Therefore, it cannot be read into SAS/IML

®
 as a numeric matrix. 

Next, we label the samples. More precisely, we replace the 0-1 variable 'sample1' by proper labels (1, 2, 3, …) for the 
samples. We use SAS/IML

®
 for this purpose: 

proc iml; 

use dhps1; 

read all into A; 

close dhps1; 

call delete(dhps1);     

B=A; 

B[,ncol(A)]=J(nrow(A),1,1); 

inc=1; 

label=1; 

do k=1 to nrow(A); 

  if A[k,ncol(A)]=0 then do; 

    inc=inc+1; 

  end; 

  else do; 

    B[k,ncol(A)]=label; 

 if inc>1 then do; 

    B[k-inc:k-1,ncol(A)]=J(inc,1,label-1); 

    inc=1; 

 end; 

 label=label+1; 

  end; 

end; 

create dhps1 from B; 

edit dhps1; 

append from B; 

quit; 

 

Figure 3 shows a screenshot of the manipulated dataset.  

Now, we eliminated the problems with the initial dataset, and are ready to rearrange the data such that the 
heterozygosity can be calculated.   

Figure 3. Screenshot of dataset „sasgf.dhps1‟ after the SAS/IML
®
 procedure. The last column was replaced, 

by the labels of the samples. (Note that the column labels are replaced.) 

 

Step2: Rearranging the data 

To calculate the heterozygosity at the various MS loci, we rearrange the dataset in a more convenient way. Namely, 
the first column specifies whether the sample carries sensitive or resistant parasites (0=sensitive, 1=resistant). The 
second column specifies the MS locus, in the third column the MS alleles are entered, and the fourth column 
specifies the weight.  

proc iml;  

use dhps1; 

read all into A; 

n=ncol(A)-2; 

m=nrow(a) ; 

B=J(m*n,4); 

do k=1 to n; 

  B[1+(k-1)*m:k*m,1]=J(m,1,k); 

  B[1+(k-1)*m:k*m,2]=A[1:m,1]; 

  B[1+(k-1)*m:k*m,3]=A[1:m,ncol(A)]; 

  B[1+(k-1)*m:k*m,4]=A[1:m,k+1]; 

end; 

create dhps2 from B; 

edit dhps2; 

append from B; 

quit; 

 

In the next step, we delete all rows with missing entries, because they are no longer needed.  The outcome is 
illustrated in Figure 4.  

data dhps2; 

set dhps2; 

if Col4=. then delete; 

run; 

Figure 4. Left: Screenshot of dataset „sasgf.dhps2‟ after performing the SAS/IML
®
 procedure and the data 

step in step 2. The labels of the blood samples are in COL3. Middle: Screenshot of dataset „sasgf.dhps2‟ after 
performing the SAS/IML

®
 procedure in step 3. COL3 was replaced by the weights. Right: Screenshot of 

dataset „dhpsfreq‟.  

 

Step3: Adding weights 

In order to correctly calculate the heterozygosity we need to weight the data (see above). Again, SAS/IML
®
 serves to 

complete this task.  In dataset „dhps2‟, we replace the column 3, which contains the labels, by a column containing 
the weights. The column containing the labels is necessary to calculate the weights but not necessary to calculate the 
heterozygosity once the weights are known. We use the following code: 

proc iml; 

use dhps2; 

read all into A; 

close dhps2; 

call delete(dhps2); /* deletes file 'work.dhps2' */ 

B=J(nrow(A),ncol(A),.); 

B[,1:ncol(A)]=A[,]; 

BB=J(nrow(A),1,1); 

inc=1; 

do k=2 to nrow(A); 

  if A[k,ncol(A)-1]=A[k-1,ncol(A)-1] then do; 

    inc=inc+1; 

  end; 

  else do; 

 if inc>1 then do; 

    BB[k-inc:k-1,1]=J(inc,1,inc); 

    inc=1; 

 end; 

  end; 

end; 

B[,ncol(A)-1]=1/BB; 

create dhps2 from B; /* creates new file 'work.dhps2' from matrix B */ 

edit dhps2; 

append from B; 

quit; 

Step4: Calculating the allele frequencies 

We need to calculate the alleles frequencies at the various MS loci to be able to derive the heterozygosity. From the 
dataset „dhps2‟, it is easy to calculate the relative frequencies of the various alleles at each MS locus among the 
samples containing resistant and sensitive parasites, respectively.  We use „proc freq‟ to perform a table analysis 
grouped by the MS loci and sensitive/resistant samples. We further export the output of the table analysis into a 
SAS

®
 dataset using SAS/ODS

®
. This dataset contains the desired frequencies to calculate the heterozygosity.  

We use the following SAS
®
 code: 

proc sort data=work.dhps2 out=work.dhps2; /*sorts the data in a proper way*/ 

 by COL2 COL1; 

run; 

 

ods listing close; 

proc freq data=work.dhps2; 

by COL2 COL1;                                                                  

    tables COL4; 

    weight COL3; 

    ods output  OneWayFreqs=dhpsfreq; 

run; 

ods listing; 

 

data work.dhpsfreq;   

set  work.dhpsfreq; 

drop Table;  /* deletes the unnecessary variable ‘Table’ */ 

run; 

 

The newly created dataset „work.dhpsfreq‟ contains the allele frequencies at the respective MS loci.  

 

Step 5: Calculating the heterozygosity 

Now, we calculate the heterozygosity at each MS locus grouped by sensitive/resistant. We use SAS/IML
®
 to create a 

dataset in which the first column specifies sensitive/resistant, the second column the MS locus, and the third column 
the value of the heterozygosity at the respective MS locus. We use the following code. 

proc iml; 

call delete(dhps2); /* deletes the now unnecessary dataset ‘dhps2’ */ 

use dhpsfreq; 

read all into A; 

B=A[1:nrow(a)-1,2]-A[2:nrow(a),2]; 

index=J(1,1,1); 

ind=1; 

do k=1 to nrow(B); 

  if B[k,1]^=0 then do; 

    ind= k+1; 

  index=index//ind; 

  end; 

end; 

ind=nrow(A)+1; 

index=index//ind; 

 

A[,5]=(A[,5]/100); 

n=nrow(index); 

C=J(n-1,3,.); 

do k=1 to n-1; 

   vec=A[index[k,1]:index[k+1,1]-1,5]; 

   vec1=vec##2; 

   m1=index[k+1,1]- index[k,1]; 

   if m1=1 then do; 

      He=0; 

   end; 

   else do; 

      He=(m1/(m1-1))*(1- vec1[+,]); 

   end; 

   C[k,1:2]=A[index[k,1],1:2]; 

   C[k,3]=He; 

end; 

create Het from C; 

edit Het; 

append from C; 

quit; 

 

Figure 5 shows a screenshot of the dataset „work.het‟.  

 

 

 

Figure 5. Screenshot of dataset „work.het‟. 

 

Step6: Plotting the results 

Finally, we can plot the dataset using „proc gplot‟.  First, we specify a new variable that will be used for the plot legend 
and specify several goptions.  

data het; 

label   col='a0'x; 

set  sasgf.het; 

if col1=0 then col="sensitive"; 

if col1=1 then col="resistant"; 

run; 

 

axis1 value=( height=1.5  angle=-30 rotate=-0  

"-72.7"  "-34.5"  "-18.7"  "-11"  "-7.4" "-2.8" "-1.5" "-0.132" "0.034" "0.5" "1.4" 

"6.4" "9" "16.3" "22.8" "36" "49.5" "66.1")  

     label=(angle=0 h=2 color=black "distance from dhps in kb" )  

     major=(number=18 height=.1 cm  ) 

     minor=( ); 

 

axis2 value=(height=1.5) label=(angle=90 h=2 color=black "Heterozygosity" )  

      major=(height=.1 cm) 

      minor=( ); 

 

LEGEND1 value=(height=1.5); 

symbol1 i =std1mt mode=include interpol=join 

        value=dot  

        ; 

symbol2 interpol=join 

        value=dot 

        ; 

 

Now, we can plot the heterozygosity. This is done with the following code. Figure 6 shows the plot.  

proc gplot data=het; 

   plot col3*col2=col / haxis=axis1  vaxis=axis2 Legend=Legend1; 

run;  

 

It is apparent from the plot that the heterozygosity among resistant parasites is reduced around the dhps gene 

compared with the heterozygosity among sensitive parasites. This is a pattern of a selective sweep. It indicates that 
selection (induced by drug pressure) is acting to increase the frequency of resistant parasites.   

 

Figure 6. Screenshot of the gplot. 

CONCLUSION 

We provided a simple example how SAS
®
 can be used to explore genetic data. We provided a rather simple and 

basic example for a genetic analysis. However, it gives some ideas how various SAS
®
 components can be combined 

to perform genetic analyses. Of course, our example can be extended to perform more comprehensive analyses of 
genetic data.  

